@echo off
:open
cls&echo ping ipv4线路
ping 8.8.8.8
if %ERRORLEVEL%==1 goto BothFail

cls&echo ping ipv6线路
ping test6.ustc.edu.cn
if %ERRORLEVEL%==0 goto Ok

:No
COLOR 74
cls&echo V6状态异常 %date%_%time%
:: 下行是网络断线后输出日志选项，删除“::”开启V6故障日志功能
::echo V6连接失败 %date%_%time%>>netconnect.txt
echo 禁用指定网络
netsh interface set interface "WINET" admin=DISABLED
ping 127.0.1 /n 5 >nul
echo 开放指定网络
netsh interface set interface "WINET" admin=ENABLED
ping 127.0.1 /n 20 >nul
goto open

:BothFail
COLOR 47
cls&echo V4状态发生异常，暂不进行网络重启 %date%_%time%
::下行是网络断线后输出日志选项，删除“::”开启V4故障日志功能
::echo V4连接失败 %date%_%time%>>netconnect.txt
ping 127.0.1 /n 60 >nul
goto open

:Ok
COLOR E2
cls&echo V6状态正常 %date%_%time%
ping 127.0.1 /n 300 >nul
goto open
::ver. 2.3.0 加入v4联合判断功能，修复不显示ping进度
::旨在为需要局域网远程连接（物理机控制困难）的终端，提供DHCP故障环境下的IP维持功能（仍能保持原有IP在局域网被控）